
ABSTRACT
While the SAS Programmer can stick strictly to SAS, where are other languages that will assist in making their job
a little easier.
This paper presents four VBA macro utilities, with code – one that will take all ASCII text files in a directory and save
them as Word files; a second will print out the first two pages of all Word files in a directory; a third will take all
Word files and concatenate these into a single Word file; and finally last but not least, a VBA macro that will search
though all the LOG files in a directory and put any issues found into a Word file for review.

INTRODUCTION
SAS Programmers occasionally have to look beyond SAS to do tasks that can be more efficiently outside SAS.
Four such tasks are:
●
●
●
●

converting files from one format to another (this paper will look at ASCII Text to Word, but other
conversions are possible with little modification)
print the first two pages of all Word documents in a directory (although this could be changed easily to
multiple pages
take all Word documents in a directory and concatenate them into one file
use a VBA macro to review all SAS LOGs in a directory and put any issues found into a single Word file for
review

VBA code will be presented in this paper that will do these tasks along with a description on how to bring these
VBA utilities into your environment so that your SAS programmers can use these also with a click of a mouse.

FIRST QUESTION, WHAT IS VBA?
The first question that comes up when talking about VBA, or it true name Visual Basic for Applications, is what is it?
In its simplest form, VBA is a programming language that is inside products like Microsoft Word that allow for the

building of user defined functions and automating processes, closely related to the programming language
Visual Basic and its predecessors, Quick Basic and WordBasic. VBA programs can be fixed to a menu button,
a keyboard shortcut, or a window icon outside the Microsoft Word environment.
Microsoft Word also has its own integrated development environment (IDE), an example of which is shown in
Display 1:

Display 1: Example of VBA IDE Environment
It is in the IDE that VBA source can be written, edited and complied for use in your work. In a later section we shall
see how to compile and attach a VBA macro to a Windows icon for running in your environment.

CONVERTING FILES FROM ONE FORMAT TO ANOTHER
Microsoft Word, from its early days had the capability to import files of different formats and save them in another
format with the most common format that a file is saved in being Microsoft Word. While this can be done quickly
done with a single file manually, when multiple files are to be converted this becomes a very tedious task and
prone to “user error”.
Using VBA, it is possible to get the list of documents in a directory and then save these in a particular format – in
this paper we shall save these as a Microsoft Word DOC format file. In the code below is a shorten version of the
code that will use default settings like line spacing, while in Appendix A there is a more complete version that
defines explicitly these settings:
01 Sub TXT2Word()
02 Dim strFileName As String
03 Dim strOutFileName As String
04 Dim strPath As String
05 With Application.FileDialog(msoFileDialogFolderPicker)
06 .InitialFileName = "C:\"
07 .Show
08 strPath = .SelectedItems(1)
09 End With
10 MsgBox "You selected " & strPath
11 strFileName = Dir(strPath + "\*.txt", vbNormal)
12 Do While strFileName <> ""
13
strOutFileName = Replace(strFileName, ".txt", ".doc")
14
Documents.Open FileName:=strPath + "\" + strFileName, ConfirmConversions:= _
15
False, Encoding:=1252
16
With ActiveDocument.Styles(wdStyleNormal).Font
17
If .NameFarEast = .NameAscii Then
18
.NameAscii = ""
19
End If

20
.NameFarEast = ""
21
End With
22
With ActiveDocument.PageSetup
23
.Orientation = wdOrientLandscape
24
.LayoutMode = wdLayoutModeDefault
25
End With
26
Selection.WholeStory
27
Selection.Font.Name = "Courier New"
28
Selection.Font.Size = 8
29
With Selection.ParagraphFormat
30
.LineSpacingRule = wdLineSpaceExactly
31
.LineSpacing = 8
32
.WordWrap = True
33
End With
34
ActiveDocument.SaveAs FileName:=strPath + "\" + strOutFileName, _
35
FileFormat:= wdFormatDocument
36
ActiveWindow.Close
37
strFileName = Dir
38 Loop
39 With Application
40
.ScreenUpdating = False
41
Do Until .Documents.Count = 0
42
'Close no save
43
.Documents(1).Close SaveChanges:=wdDoNotSaveChanges
44
Loop
45
.Quit SaveChanges:=wdDoNotSaveChanges
46 End With
47 End Sub
In line 5, the user is presented with a dialog box, similar to Display 2, where the user can select the directory to
process – note that all the files in the directory with the extension TXT will be selected to run using the code above
and in Appendix A.

Display 2: Selection of directory to process
Line 10 will display the folder selected in a standard Windows Message Box, and line 11 will hold a list of the files
to internally process. Lines 12 to 38 open the document, set orientation to landscape, set font and size of font,
then finally save the document in Word DOC format. Line 39 to 46 just make certain all files are closed before the
macro finishes.

PRINTING THE FIRST TWO PAGES OF ALL DOC FILES IN A DIRECTORY
It is sometimes necessary to print out the first few pages of all the files of a certain format in a directory but not print
out all pages. A VBA macro that will take all the DOC format files in a directory and print out to the default printer
only the first two pages of each file is given below:
Sub PrintPgsWord()
Dim strFileName As String
Dim strPath As String
With Application.FileDialog(msoFileDialogFolderPicker)
.InitialFileName = "C:\"
.Show
strPath = .SelectedItems(1)
End With
MsgBox "You selected " & strPath
strFileName = Dir(strPath + "\*.doc", vbNormal) ' can be .rtf or .docx
Do While strFileName <> ""
Documents.Open FileName:=strPath + strFileName
Documents(strFileName).PrintOut Range:=wdPrintFromTo, From:="1", To:="2"
Documents.Close
strFileName = Dir
Loop
End Sub
4

Four Useful VBA Utilities for SAS Programmers, continued

When this macro is run a selection display similar to Display 2 is shown where the user has the ability to select a
directory to process. All DOC files, specified in line 10, are processed, with the page range setting being set in line
that reads “From:="1", To:="2"”.

CONCATENATE ALL WORD DOCUMENTS IN A DIRECTORY INTO A SINGLE WORD
FILE
The VBA macro shown here will take all the RTF format files in a directory and put them into single Word file based
on the file name order that Windows uses. Note that if your want to “fudge” the order, i.e. put one file in before
another, the only way is to change the name of the document itself. Although this macro uses the RTF format files,
it will work the same with .DOC or DOCX files. Note that this macro uses a set directory to read the individual files
to be combined, i.e. D:\TEMP\RTF, this could be altered for the user to select the directory as in the previous
examples.
Sub MergeWrdDocs()
Dim RngA As Range
Dim CombDoc As Document
Dim FileInDoc As String
Const FolderLoc = "D:\TEMP\RTF\" 'change to your own situation
Set CombDoc = Documents.Add
FileInDoc = Dir$(FolderLoc & "*.rtf") ' can be .doc or .docx
Do Until FileInDoc = ""
Set RngA = CombDoc.Range
RngA.Collapse wdCollapseEnd
RngA.InsertFile FolderLoc & FileInDoc
RngA.InsertBreak Type:=wdSectionBreakNextPage
FileInDoc = Dir$()
Loop
ActiveDocument.SaveAs (FolderLoc + "COMBINED.doc")
End Sub
In this macro the output file name is set to COMBINED.DOC however this can be changed in line 10 to any other
name. Note also, that if you page numbers in the combined document are not changed from the original
document.

REVIEW ALL SAS LOGS IN A DIRECTORY
I have stated in previous papers, when QCing your SAS program it is extremely important to review the SAS LOG. I
have also presented a SAS macro that will do this task but it is also possible to do this in a VBA – in these days of
cutbacks a QCer of your program may not have SAS but has very likely Microsoft Word on their computer.
Below is a VBA macro that will review all the SAS LOGs in the chosen directory and put any findings into a
document called QCFINDINGS.DOC.
Note, as in the prioir macro, that this macro uses a set directory to read the individual files to be combined, i.e.
C:\TEMP\LOGS, this could be altered for the user to select the directory as in the previous examples.
Sub QCLOGER()
Dim sFileName As String
Dim iFileNum As Integer
Dim sBuf As String
Dim Fields As String
Dim TempStr As String
Const FolderLoc = "C:\TEMP\LOGS\" 'change to your situation
ChDir FolderLoc

Set CombDoc = Documents.Add
' Header for file
Selection.InsertAfter ("QC Log Check for Directory ") & FolderLoc & vbCrLf
Selection.InsertAfter (" ") & vbCrLf
Selection.InsertAfter (" ") & vbCrLf
sFileName = Dir$(FolderLoc & "*.LOG") ' Load all LOG files
Do Until sFileName = ""
' Header for each file
Selection.InsertAfter ("===========================") & vbCrLf
Selection.InsertAfter ("Log Name: ") & sFileName & vbCrLf
Selection.InsertAfter (" ") & vbCrLf
Open sFileName For Input As #1
k=0
Do While Not EOF(1)
Line Input #1, Fields
lPA = InStr(1, Fields, "ERROR:") ' ERROR text position in line
lPB = InStr(1, Fields, "WARNING:") ' WARNING text position in line
If lPA > 0 Or lPB > 0 Then ' One of the messages found
Selection.InsertAfter Fields & vbCrLf
k=k+1
End If
Loop
If k = 0 Then ' No issues found, report
Selection.InsertAfter ("***No Issues Found***") & vbCrLf
End If
Selection.InsertAfter (" ") & vbCrLf
Close #1
sFileName = Dir$() ' Get next file
Loop
Selection.InsertAfter (" ") & vbCrLf ' Close out file
Selection.InsertAfter ("/*EOF*/") & vbCrLf
ActiveDocument.SaveAs (FolderLoc + "QCFINDINGS.DOC") ' Save file
End Sub
Note that the selection of the directory to process is the same as in Display 2. In the VBA code provided, the
search text that is chosen is “ERROR:” and “WARNING”, line block 1214, but this can be extended to other search
terms as needed in your local environment by editing this file. Also note that should no findings be found, a
message stating that no findings were found is put into the document. Below, in Display 3, is a screen shot of an
example file.

CONCLUSION
This paper has looked into four VBA utilities that are useful for the SAS programmer four their work. It is possible
to do all these tasks within SAS, but it is not always the best way to do these – sometimes the SAS programmer
has to look beyond SAS to do his/her work.

Appendix A: Listing of TXT2Word VBA Macro
Sub TXT2Word()
' TXT2Word Macro
Dim strFileName As String
Dim strOutFileName As String
Dim strPath As String
With Application.FileDialog(msoFileDialogFolderPicker)
.InitialFileName = "C:\"
.Show
strPath = .SelectedItems(1)
End With
MsgBox "You selected " & strPath
strFileName = Dir(strPath + "\*.txt", vbNormal)
Do While strFileName <> ""
strOutFileName = Replace(strFileName, ".rpt", ".doc")
strOutFileName = Replace(strOutFileName, ".RPT", ".doc")
Documents.Open FileName:=strPath + "\" + strFileName, ConfirmConversions:= _
False, ReadOnly:=False, AddToRecentFiles:=False, PasswordDocument:="", _
PasswordTemplate:="", Revert:=False, WritePasswordDocument:="", _
WritePasswordTemplate:="", Format:=wdOpenFormatAuto, XMLTransform:="", _
Encoding:=1252
With ActiveDocument.Styles(wdStyleNormal).Font
If .NameFarEast = .NameAscii Then
.NameAscii = ""
End If
.NameFarEast = ""
End With
With ActiveDocument.PageSetup
.LineNumbering.Active = False
.Orientation = wdOrientLandscape
.TopMargin = InchesToPoints(1)
.BottomMargin = InchesToPoints(1)
.LeftMargin = InchesToPoints(1)
.RightMargin = InchesToPoints(1)
.Gutter = InchesToPoints(0)
.HeaderDistance = InchesToPoints(0.5)
.FooterDistance = InchesToPoints(0.5)
.PageWidth = InchesToPoints(11)
.PageHeight = InchesToPoints(8.5)
.FirstPageTray = wdPrinterDefaultBin
.OtherPagesTray = wdPrinterDefaultBin
.SectionStart = wdSectionNewPage
.OddAndEvenPagesHeaderFooter = False
.DifferentFirstPageHeaderFooter = False
.VerticalAlignment = wdAlignVerticalTop
.SuppressEndnotes = False
.MirrorMargins = False
.TwoPagesOnOne = False
.BookFoldPrinting = False
.BookFoldRevPrinting = False
.BookFoldPrintingSheets = 1
.GutterPos = wdGutterPosLeft

.LayoutMode = wdLayoutModeDefault
End With
Selection.WholeStory
Selection.Font.Name = "Courier New"
Selection.Font.Size = 8
With Selection.ParagraphFormat
.SpaceBeforeAuto = False
.SpaceAfterAuto = False
.LineSpacingRule = wdLineSpaceExactly
.LineSpacing = 8
.WordWrap = True
End With
ActiveDocument.SaveAs FileName:=strPath + "\" + strOutFileName, FileFormat:= _
wdFormatDocument, LockComments:=False, Password:="", AddToRecentFiles:= _
True, WritePassword:="", ReadOnlyRecommended:=False, EmbedTrueTypeFonts:= _
False, SaveNativePictureFormat:=False, SaveFormsData:=False, _
SaveAsAOCELetter:=False
ActiveWindow.Close
strFileName = Dir
Loop
With Application
.ScreenUpdating = False
'Loop Through open documents
Do Until .Documents.Count = 0
'Close no save
.Documents(1).Close SaveChanges:=wdDoNotSaveChanges
Loop
'Quit Word no save
.Quit SaveChanges:=wdDoNotSaveChanges
End With
End Sub

